JAPLA研究会資料 2014/4/19

Ｊによるエルミット多項式の導出と調和振動子の量子力学

西川　利男

１．調和振動とエルミットの方程式[1]

物理学においては、調和振動として近似される系はたくさんある。最も簡単にはフックの法則に従うバネに結ばれた質点の運動であり、ミクロの世界では２原子分子の振動や結晶格子内の原子の振動がある。

調和振動子というのは質量 m の粒子がポテンシャル

$$V = \frac{1}{2}kx^2$$

のもとで力を受けて、一直線上を x 軸に沿って動いているものである。粒子の受ける力は $-kx$ であるから、運動方程式は古典的には

$$m\frac{d^2x}{dt^2} = -kx$$

であって、一般解は次のようになる。

$$x = a\cos 2\pi\nu(t - t_0)$$ 　a と t_0 とは定数で　$\nu = \frac{1}{2\pi}\sqrt{\frac{k}{m}}$　である。

この運動エネルギー　$\frac{1}{2}m\left(\frac{dx}{dt}\right)^2$　と位置エネルギー $\frac{1}{2}kx^2$　から

古典的ハミルトニアンは

$$H = \frac{1}{2m}p^2 + \frac{k}{2}x^2$$

となり、ハミルトン演算子は

$$H = -\frac{h^2}{8\pi^2 m}\frac{d^2}{dx^2} + \frac{k}{2}x^2$$

となり、この系の波動方程式は

$$\frac{d^2\psi}{dx^2} + \frac{8\pi^2 m}{h^2}\left(E - \frac{k}{2}x^2\right)\psi = 0$$

である。

これから、いくつかの式の置き換えを行った後、波動関数は次のエルミットの微分方程式を解くことで得られることになる。

$$\frac{d^2u}{dx^2} - 2x\frac{du}{dx} + 2nu = 0$$

この解は

$$H_n(x) = (-1)^n e^{x^2}\frac{d^n}{dx^n}(e^{-x^2})$$

なるエルミットの多項式で表される。

[1] アイリング、ウォルター、キンボール、小谷正雄訳「量子化学」p.86-90
山口書店　(1953).

２．エルミットの多項式のＪによる算出

エルミットの多項式から次の漸化式が得られる。

$$H_n(x) = 2xH_{n-1}(x) - 2(n-1)H_{n-2}(x)$$

はじめの数項は次のようになる。

$$H_0(x) = 1$$

$$H_1(x) = 2x$$

$$H_2(x) = 4x^2 - 2$$

$$H_3(x) = 8x^3 - 12x$$

$$H_4(x) = 16x^4 - 48x^2 + 12$$

これをもとにしてＪにより次数ｎまでのエルミットの多項式の係数を算出するプログラムを作った。

ここでは、上とは異なり、昇べきの順で多項式として係数を算出している。
まず、$n = 0$ と $n = 1$ とについては、直接結果を与え、その後は、以前の値を使って次々と漸化式を使って算出する。

```
NB. Hermite Polynomial
NB.   programmed by T.N. 2014/3/23
wr =: 1!:2&2

herm =: 3 : 0
n =. y.
H =. (0, 1)$ ''
i =. 0
while. i <: n
  do.
    wr 'i:', ": i
    if. i = 0 do. Hx =. (1, 1)$1 end.
    if. i = 1 do. Hx =. 0, 2  end.
    if. i > 1 do.
       Hx0 =. 0, 2 * (i-1) { H
       wr 'Hx0:', ":Hx0
       Hx1 =. (_2 * (i-1) * (i-2){ H) , 0
       wr 'Hx1:', ":Hx1
       Hx =. Hx0 + Hx1
      end.
    H =. H, Hx
    i =. i + 1
    wr '--------'
  end.
wr 'Hermite Polynomial H: ============='
H
)
```

実行した結果は次のようになる。

```
 herm 6
i:0
--------
i:1
--------
i:2
Hx0:0 0 4
Hx1:_2 0 0
--------
i:3
Hx0:0 _4 0 8
Hx1:0 _8 0 0
--------
i:4
Hx0:0 0 _24 0 16
Hx1:12 0 _24 0 0
--------
i:5
Hx0:0 24 0 _96 0 32
Hx1:0 96 0 _64 0 0
--------
i:6
Hx0:0 0 240 0 _320 0 64
Hx1:_120 0 480 0 _160 0 0
--------
Hermite Polynomial H: =============
   1   0   0    0    0  0  0
   0   2   0    0    0  0  0
  _2   0   4    0    0  0  0
   0 _12   0    8    0  0  0
  12   0 _48    0   16  0  0
   0 120   0 _160    0 32  0
_120   0 720    0 _480  0 64
```

３．調和振動子の波動関数のグラフ表示

調和振動子のエネルギー準位はつぎのようになる。

$$E_n = \left(n + \frac{1}{2} \right) h\nu$$

また、エネルギー準位のそれぞれに対応した波動関数はつぎのようになる。

$$\psi_n(x) = \left(\frac{K}{2^2 n!} \right)^{1/2} H_n(x)\, e^{-1/2\, x^2}$$

　これから、調和振動子の各エネルギー準位に対応した波動関数をＪのグラフィックスにより図示しようと思う。
　これにはＪの plot ルーチンで簡単に行うことができる。プログラムは以下のとおりである。

```
NB. plot ================================
load 'numeric'
load 'plot'

test =: 3 : 0
X =. range _3 3 0.1
HX =. _2 + 4 * X^2    NB. Hermite
EX =. ^ - -: X^2      NB. exp(-1/2 * X^2)
Y =. HX * EX
plot X;Y
)

coef =: 3 : '((2^y.)*(! y.))^(-1r2)'

run =: 3 : 0
X =. range _3.5 3.5 0.1

H0X =. 1
H1X =. 2 * X
H2X =. _2 + 4 * X^2
H3X =. (_12*X) + (8*X^3)
H4X =. 12 + (_48*X^2) + (16*X^4)
H5X =. (120*X) + (_160*X^3) + (32*X^5)

EX =.  ^ - -: X^2

Y0 =. 1r2 + 0.5*(coef 0)*H0X * EX
Y1 =. 3r2 + 0.5*(coef 1)*H1X * EX
Y2 =. 5r2 + 0.5*(coef 2)*H2X * EX
Y3 =. 7r2 + 0.5*(coef 3)*H3X * EX
Y4 =. 9r2 + 0.5*(coef 4)*H4X * EX
Y5 =. 11r2 + 0.5*(coef 5)*H5X * EX

Z  =. 1r2 * X^2

Y =.  Z, Y5, Y4, Y3, Y2, Y1,: Y0

plot X;Y
)
```

実行したようすは次のとおりである。